Szereg Fouriera-Legendre'a

Szereg Fouriera-Legendre'a : $\sum_{n=0}^{\infty} P_n(t)\, f_n$

Współczynniki $f_n = \frac{\int_{-1}^{1} P_n(t)\, f(t)\, dt}{\int_{-1}^{1} P_n(t)^2\, dt} = \frac{2\,n+1}{2} \int_{-1}^{1} P_n(t)\, f(t)\, dt$

$\int_{-1}^{1} P_n(t)^2\, dt = \frac{2}{2\,n+1}$

Zadanie 1

Policz kwadrat normy funkcji Legendre'a (LegendreP[n,t])

$\int_{-1}^{1} P_n(t)^2\, dt$
dla n=0, 1, 2, 3

Funkcja zwracajaca współczynniki Fouriera-Legendre'a

Zdefiniujmy funkcję zwracajaca współczynniki Fouriera-Legendre'a $f_n = \frac{2\,n+1}{2} \int_{-1}^{1} P_n(t)\, f(t)\, dt$:

```
FourierLegendre[f_, t_, n_] := (2 n + 1) / 2 Integrate[LegendreP[n, t] * f, {t, -1, 1}]
```

Przykład

Exp[$-t$]

```
f[t_] = Exp[-t]
```

e^{-t}

Lista 4 pierwszych współczynników Fouriera-Legendre'a:

```
fn = Table[FourierLegendre[f[t], t, n], {n, 0, 3}]
```

$$\left\{\sinh(1), -\frac{3}{e}, \frac{5}{2}\left(e-\frac{7}{e}\right), \frac{7}{2}\left(5\,e-\frac{37}{e}\right)\right\}$$

Lista 4 pierwszych funkcji Legendre'a:

```
L = Table[LegendreP[n, t], {n, 0, 3}]
```

$$\left\{1, t, \frac{1}{2}(3\,t^2-1), \frac{1}{2}(5\,t^3-3\,t)\right\}$$

Suma cześciowa szeregu do n=3 $\sum_{n=0}^{3} P_n(t)\, f_n$ wynosi:

```
fn.L
```

$$\frac{7}{4}\left(5\,e-\frac{37}{e}\right)(5\,t^3-3\,t)+\frac{5}{4}\left(e-\frac{7}{e}\right)(3\,t^2-1)-\frac{3\,t}{e}+\sinh(1)$$

Funkcja zwracajaca sumy czesciowe szeregu $\sum_{n=0}^{k} P_n(t)\, f_n$ od k=0 do k=K

```
WypiszSumyCzesciowe[funkcja_, K_] :=
 Module[{Wynik, SumaCzesciowa, L, fn},  (* zmienne lokalne*)

  fn = Table[FourierLegendre[funkcja, t, n], {n, 0, K}];
  (* Lista K+1 pierwszych współczynników Fouriera-Legendre'a *)

  L = Table[LegendreP[n, t], {n, 0, K}];
  (* Lista K+1 pierwszych  funkcji Legendre'a *)

  Wynik = {};  (* pusta lista *)
  Do[SumaCzesciowa = (Take[L, n]).(Take[fn, n]);
   Wynik = Append[Wynik, SumaCzesciowa], {n, 1, Length[L]}];
  (* Tworzy sumy czesciowe i dodaje je do  wyniku*)
  Wynik
 ]
```

Moduł

```
?Module
```

Module[{x, y, …}, *expr*] specifies that occurrences of the symbols x, y, … in *expr* should be treated as local.
Module[{$x = x_0$, …}, *expr*] defines initial values for x, …. >>

Take

Take[lista,n] zwraca pierwszych n elementów listy

```
? Take
```

Take[*list*, *n*] gives the first *n* elements of *list*.
Take[*list*, −*n*] gives the last *n* elements of *list*.
Take[*list*, {*m*, *n*}] gives elements *m* through *n* of *list*.
Take[*list*, seq_1, seq_2, …] gives a nested list in which elements specified by seq_i are taken at level *i* in *list*. >>

Append

Append[lista, elem] zwraca listę powiekszoną o element elem

```
? Append
```

Append[*expr*, *elem*] gives *expr* with *elem* appended.
Append[*elem*] represents an operator form of Append that can be applied to an expression. >>

Przykład

Wykres przedstawiajacy funkcję f oraz sumy czesciowej jej rozwiniecia w szereg Fouriera-Legendre'a n=0,1,2,3

```
WypiszSumyCzesciowe[f[t], 3]
```

$$\left\{\sinh(1),\ \sinh(1)-\frac{3\,t}{e},\ \frac{5}{4}\left(e-\frac{7}{e}\right)\left(3\,t^2-1\right)-\frac{3\,t}{e}+\sinh(1),\ \frac{7}{4}\left(5\,e-\frac{37}{e}\right)\left(5\,t^3-3\,t\right)+\frac{5}{4}\left(e-\frac{7}{e}\right)\left(3\,t^2-1\right)-\frac{3\,t}{e}+\sinh(1)\right\}$$

```
Plot[Evaluate[{f[t], WypiszSumyCzesciowe[f[t], 3]}], {t, -1.2, 1.2},
 PlotRange → {-2, 3},
 PlotStyle → {{Thickness[.01], Black}, Yellow, Orange, Red, Green},
 PlotLegends → "Expressions"]
```



Zadanie 2

Narysuj wykres przedstawiajacy funkcję f oraz sumy czesciowej jej rozwiniecia w szereg Fouriera-Legendre’a n=0,1,2,3,4,5 dla

a)f(t)= sinh(t+|t|)

b)f(t)= Π (t - 0.25)   (funkcja HeavisidePi)

Zadanie 3

Utworz funkcję

WypiszSumyCzescioweParzyste[funkcja_,K_]

ktora bedzie tworzyc sumy parzystych wyrazów szeregu Fouriera-Legendre’a

($\{f_0 P_0,\ f_0 P_0 + f_2 P_2,\ f_0 P_0 + f_2 P_2 + f_4 P_4, \ldots\}$)

Narysuj wykresy przedstawiajace funkcję f oraz sumy czesciowe jej rozwiniecia w szereg Fouriera-Legendre’a

a) cos(π t)

b) exp(-|t|)

Trygonometryczny Szereg Fouriera

FourierTrigSeries[expr, t, n]

$a_0/2+\sum_{k=1}^{n} a_k \cos(k\,t)+b_k \sin(k\,t)$

$a_k=\frac{1}{\pi}\int_{-\pi}^{\pi} f(t)\cos(k\,t)\,dt$

$b_k=\frac{1}{\pi}\int_{-\pi}^{\pi} f(t)\sin(k\,t)\,dt$

FourierTrigSeries

Wypisuje rozwiniecie funkcji w szerg Fouriera do n-tego wyrazu

```
? FourierTrigSeries
```

FourierTrigSeries[*expr*, *t*, *n*] gives the n^{th}-order Fourier trigonometric series expansion of *expr* in *t*.
FourierTrigSeries[*expr*, {t_1, t_2, …}, {n_1, n_2, …}] gives the multidimensional Fourier trigonometric series of *expr*. >>

Przykład

$f(t)=t+t^2$

t∈$(-\pi,\pi)$

```
fa[t_] = t + t^2
```

t^2+t

```
FourierTrigSeries[fa[t], t, 5]
```

$$-2\left(-\sin(t)+\frac{1}{2}\sin(2\,t)-\frac{1}{3}\sin(3\,t)+\frac{1}{4}\sin(4\,t)-\frac{1}{5}\sin(5\,t)\right)+4\left(-\cos(t)+\frac{1}{4}\cos(2\,t)-\frac{1}{9}\cos(3\,t)+\frac{1}{16}\cos(4\,t)-\frac{1}{25}\cos(5\,t)\right)+\frac{\pi^2}{3}$$

```
Plot[Evaluate[FourierTrigSeries[fa[t], t, 5]], {t, -1.5 Pi, 1.5 Pi}]
```



FourierParameters

FourierParameters→{a,b}
domyslnie jest {1,1}

$$\left|\frac{b}{\pi}\right|^{\frac{1-a}{2}}\left(a_0/2+\sum_{k=1}^{n} a_k\cos(b\,k\,t)+b_k\sin(b\,k\,t)\right)$$
$$a_k=\left|\frac{b}{\pi}\right|^{\frac{a+1}{2}}\int_{-\pi/|b|}^{\pi/|b|} f(t)\cos(b\,k\,t)\,dt$$
$$b_k=\left|\frac{b}{\pi}\right|^{\frac{1+a}{2}}\int_{-\pi/|b|}^{\pi/|b|} f(t)\sin(b\,k\,t)\,dt.$$

Zmiana granic przedziału
$f(t)=t+t^2$
t∈(-1,1)

FourierParameters→{1,Pi}

```
FourierTrigSeries[fa[t], t, 5, FourierParameters → {1, Pi}]
```

$$\frac{2\sin(\pi t)}{\pi}-\frac{\sin(2\pi t)}{\pi}+\frac{2\sin(3\pi t)}{3\pi}-\frac{\sin(4\pi t)}{2\pi}+\frac{2\sin(5\pi t)}{5\pi}-\frac{4\cos(\pi t)}{\pi^2}+\frac{\cos(2\pi t)}{\pi^2}-\frac{4\cos(3\pi t)}{9\pi^2}+\frac{\cos(4\pi t)}{4\pi^2}-\frac{4\cos(5\pi t)}{25\pi^2}+\frac{1}{3}$$

```
Plot[Evaluate[FourierTrigSeries[fa[t], t, 5, FourierParameters → {1, Pi}]],
 {t, -1.5 Pi, 1.5 Pi}]
```




```
WypiszSumyCzescioweTrig[funkcja_, K_, T_] := (* dla przedziału (-T,T) *)
 Module[{Wynik},
  Wynik = {};
  Do[Wynik = Append[Wynik,
     FourierTrigSeries[funkcja, t, n, FourierParameters → {1, Pi / T} ]], {n, 0, K}];
  Wynik
 ]
```

```
Plot[Evaluate[{ fa[t], WypiszSumyCzescioweTrig[fa[t], 4, Pi]}],
 {t, -Pi, Pi}, PlotRange → {-2, 4},
 PlotStyle → {{Thickness[.01], Black}, Yellow, Orange, Red, Green, Blue},
 PlotLegends → "Expressions"]
```



Zadanie 4

Narysuj wykres przedstawiajacy funkcję f oraz sumy czesciowej jej rozwiniecia wszereg Fouriera n=0,1,2,3,4 dla

a)f(t)= exp(-|t|)   dla przedziału $(-\pi,\pi)$

b)f(t)= sin(1-|t|)   dla przedziału (-1,1)

c)f(t)= Π(5*t)   dla przedziału (-1,1)

Eksponencjalny Szereg Fouriera

FourierSeries[expr, t, n]

$\sum_{k=-n}^{n} f_k \exp(k\,t)$

$f_k = \frac{1}{2\pi} \int_{-\pi}^{\pi} f(t) \exp(k\,t)\, dt$

```
? FourierSeries
```

FourierSeries[*expr*, *t*, *n*] gives the n^{th}-order Fourier series expansion of *expr* in *t*.

FourierSeries[*expr*, {t_1, t_2, ...}, {n_1, n_2, ...}] gives the multidimensional Fourier series. >>

Przykład

```
F[t_] = t^2
```

t^2

```
FourierSeries[F[t], t, 5]
```

$$-2\,e^{-i t} - 2\,e^{i t} + \frac{1}{2}\,e^{-2 i t} + \frac{1}{2}\,e^{2 i t} - \frac{2}{9}\,e^{-3 i t} - \frac{2}{9}\,e^{3 i t} + \frac{1}{8}\,e^{-4 i t} + \frac{1}{8}\,e^{4 i t} - \frac{2}{25}\,e^{-5 i t} - \frac{2}{25}\,e^{5 i t} + \frac{\pi^2}{3}$$

Transformacja Fouriera

FourierTransform[expr, t, z]

$$\frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\infty} f(t)\, e^{i z t}\, dt$$

InverseFourierTransform[expr, z, t]

$$\frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\infty} f(z)\, e^{-i z t}\, \mathrm{dz}$$

Przykład

```
FourierTransform[F[t], t, z]
```

$-\sqrt{2\pi}\, \delta''(z)$

```
InverseFourierTransform[%, z, t]
```

t^2

Zadanie 5

Policz transformacje Fouriera funkcji

a) exp(-|t|)
b) exp(- t^2)
c) Π(t)
d) δ(t)
e) 1
f) x^5
g) $\frac{1}{\sqrt{|t|}}$
h) sign(t)
i) sin(t)
j) $\sin(t^2)$
k) $\frac{1}{t}$
l) $\frac{d^2}{\mathrm{dt}^2} u(t)$

Zadanie 6

Aby rozwiazac rownanie dyfuzji

$\frac{\partial}{\partial t} f(x, t) - \frac{\partial^2}{\partial x^2} f(x, t) = 0$ z warunkiem poczatkowym f(x,0)=δ(x) rozwiaż rownanie będące jego transformacja Fouriera:

$\frac{\partial}{\partial t} u(k, t) + k^2 u(k, t) = 0$

gdzie u(k,t) = $\frac{1}{\sqrt{2\pi}} \int_{-\infty}^{\infty} f(x, t)\, e^{i k x}\, dx$
(znajdz warunek poczatkowy u(k,0) dla funkcji u(k,t))

Policz odwrotna transformate Fouriera rozwiazania.

Narysuj Plot3D rozwiazania dla {x,-10,10} oraz {t,0,10}